
	Reason for change:
	As per the Step 14b of the expected test sequence (34.123-1 Sec 11.3.2a.4) the SS releases the RB20 and RB22.
In the current TTCN implementation, the RB Release message for RB22 is incorrect. This needs to be corrected based on the RB Setup for RB22.

Also, the function used to release RB22 is not correct : It release the RL for RB22 straightaway which creates problems later in postamble.

All these issues need to be fixed.

	
	

	Summary of change:
	Created a new template cds_RadioBearerRelease_IEs_r7_11_3_2a and a new function f_UTRAN34_ReleaseAM_RAB22_r7_NISPC.
Please see detailed change description below.

	Other comments:
	


1.1 cds_RadioBearerRelease_IEs_r7_11_3_2a
	Object name 
	cds_RadioBearerRelease_IEs_r7_11_3_2a

	Reason for change
	In the current TTCN implementation, fl_TC_11_3_2a_Body()->f_UTRAN34_ReleaseAM_RAB22_r7-> cs_RB_Rel_AM22_PS_r7-> cds_RadioBearerRelease_IEs_r7. This template sets the puncturing limit incorrectly to p10_96. In the RB setup for RB22 this is set to P10_92. This needs to be corrected. Also, the DL TFCS being set are incorrect. This needs to be corrected.

	Summary of change
	Created a new template cds_RadioBearerRelease_IEs_r7_11_3_2a and used it instead of cds_RadioBearerRelease_IEs_r7. Also created a new template cs_RB_Rel_AM22_PS_r7_11_3_2a which uses the newly created template cds_RadioBearerRelease_IEs_r7_11_3_2a. This now sets the puncturing limit correctly, and the correct DL TFCS are passed as parameter to this template 

Please see screenshot.

	Module
	UTRAN34_NISPC_Templates.ttcn

	Label
	


	template (value) RadioBearerRelease_r7_IEs cds_RadioBearerRelease_IEs_r7_11_3_2a(ActivationTime              p_ActTime,
                                                                           SS_RB_Identity              p_RB_Id,
                                                                           template (value) DL_CommonTransChInfo_r4 p_DL_CommonTransChInfo,
                                                                           FrequencyInfo               p_FreqInfo,
                                                                           PrimaryScramblingCode       p_PrimScramblingCode,
                                                                           UL_ScramblingCode           p_UL_ScramblingCode)
    modifies cs_RadioBearerRelease_r7_IEs_Omit :=
  { /* @status    APPROVED (UTRAN) */
    activationTime := p_ActTime,
    ul_CommonTransChInfo := cs_UL_CommTrChInfoDCH_PS_64k_FDD,
    ul_AddReconfTransChInfoList := cs_UL_AddReconfTransChInfoListDCH_PS_64k_r7,
    dl_CommonTransChInfo := p_DL_CommonTransChInfo,
    dl_AddReconfTransChInfoList := cs_DL_AddReconfTransChInfoListDCH_PS_64k,
    frequencyInfo := p_FreqInfo,
    ul_DPCH_Info := cs_UL_DPCH_Info_FDDr7(tsc_UL_DPDCH_SF_64k_PS, pl0_92, p_UL_ScramblingCode),
    dl_CommonInformation := cs_DL_CommonInformationRB_SetUp_FDD_r7(tsc_DL_DPCH1_SFP_64k_PS),
    dl_InformationPerRL_List := cs_DL_InformationPerRL_r7_FDD (p_PrimScramblingCode, tsc_DL_DPCH1_ChC_64k_PS, tsc_DL_DPCH1_2ndScrC)
  };


	template (value) DL_DCCH_Message cs_RB_Rel_AM22_PS_r7_11_3_2a(IntegrityCheckInfo          p_IntegrityCheckInfo,
                                                        ActivationTime              p_ActTime,
                                                        FrequencyInfo               p_FreqInfo,
                                                        PrimaryScramblingCode       p_PrimScramblingCode,
                                                        UL_ScramblingCode           p_UL_ScramblingCode) :=
  /* @status    APPROVED (UTRAN) */
   cs_RadioBearerRelease_r7(p_IntegrityCheckInfo, cds_RadioBearerRelease_IEs_r7_11_3_2a(p_ActTime, tsc_RB22, cs_DL_CommonTransChInfoDCH_FDD(cs_TFCS_Cmpl0_1_2_3_4_5_6_7_8_9_Rx), p_FreqInfo, p_PrimScramblingCode, p_UL_ScramblingCode));


1.2 f_UTRAN34_ReleaseAM_RAB22_r7_NISPC
	Object name 
	f_UTRAN34_ReleaseAM_RAB22_r7_NISPC

	Reason for change
	In the current TTCN implementation, fl_TC_11_3_2a_Body()->f_UTRAN34_ReleaseAM_RAB22_r7 is being used. This function releases the RL for RB22, which can create issues later in Postamble (which also tries to release resources for RB22). This function also uses the template Also, this function is used by testcase 11.1.5.2, which needs the RB22 RL to be released within the function.

	Summary of change
	To resolve this conflict, another function f_UTRAN34_ReleaseAM_RAB22_r7_NISPC is created in which RL for RB22 is not released. This new function is used instead of f_UTRAN34_ReleaseAM_RAB22_r7. This function also uses the correct template cs_RB_Rel_AM22_PS_r7_11_3_2a for RB Release (Please refer change above).
Please see screenshot.

	Module
	UTRAN34_NISPC_Functions.ttcn

	Label
	


	function f_UTRAN34_ReleaseAM_RAB22_r7_NISPC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  {
    var UTRAN_SecurityInfo_Type v_SecurityInfo;
    var RRC_TransactionIdentifier v_RRC_Ti := tsc_RRC_TI;
    var UTRAN_CFN_Info_Type v_CFN_Info;
    var UL_ScramblingCode v_UL_ScrCode := f_UTRAN_CellInfo_GetULScrmCode_FDD(p_CellId);
    var PrimaryScramblingCode v_PrimScrCode := f_UTRAN_CellInfo_GetPriScrmCode_FDD(p_CellId);
    var FrequencyInfo v_FreqInfo := f_UTRAN_CellInfo_GetFrequencyInfo(p_CellId);
    v_CFN_Info := f_CalculateActTime(p_CellId);
    v_SecurityInfo := f_UTRAN_Security_Get();
    //@sic R5s140400 sic@
    U_AM.send(cas_RLC_Data_Req_Cnf(utran_CellDedicated, tsc_RB2, tsc_Mui, cs_RB_Rel_AM22_PS_r7_11_3_2a(v_SecurityInfo.dl_IntegrityCheckInfo,
                                                                                               v_CFN_Info.actTime,
                                                                                               v_FreqInfo,
                                                                                               v_PrimScrCode,
                                                                                               v_UL_ScrCode)));
    U_AM.receive(car_AM_DataMuiCnf(utran_CellDedicated,tsc_RB2, tsc_Mui));  //@sic R5s140400 sic@        
    f_UTRAN34_SS_2DCH_Modify_FDD_r7(p_CellId,
                                    cs_CphyTrchConfigReq_UL_DCH_336_148_FDD(cs_ActivateCFN(v_CFN_Info.actTime)),
                                    cs_CphyTrchConfigReq_DL_DCH_336_148_FDD(cs_ActivateCFN(v_CFN_Info.actTime)),
                                    cs_TrChInfoUL_336_148_FDD_NISPC, //@sic R5s150002 sic@
                                    cs_TrChInfoDL_336_148_FDD_NISPC, //@sic R5s150002 sic@
                                    cs_TrLogMappingUL_4DCCH_1DTCH(tsc_RB20),
                                    cs_TrLogMappingDL_4DCCH_1DTCH_PS,
                                    v_CFN_Info.actTime,
                                    cs_DL_DPCH_64k_PS_FDDr7(cs_DL_CommonInformationRB_SetUp_FDD_r7(tsc_DL_DPCH1_SFP_64k_PS),
                                                            tsc_DL_DPCH1_2ndScrC), //@sic R5s150002 sic@
                                    cs_UL_DPCH_Info_FDDr7(tsc_UL_DPDCH_SF_64k_PS, pl0_92, v_UL_ScrCode));    
    f_UTRAN_CellInfo_SetConfigType(p_CellId, cell_DCH_64kPS_AM_RAB);
    //Receive RB Release Complete message
    U_AM.receive(car_RLC_Data_Ind(utran_CellDedicated, tsc_RB2, cr_108_radioBearerReleaseComplete(v_RRC_Ti, *, *)));
  }
	


1.3 fl_TC_11_3_2a_Body()
	Object name 
	fl_TC_11_3_2a_Body()

	Reason for change
	In the current TTCN implementation, fl_TC_11_3_2a_Body() calls the function f_UTRAN34_ReleaseAM_RAB22_r7. Replaced this with f_UTRAN34_ReleaseAM_RAB22_r7_NISPC (Please refer to changes above)

	Summary of change
	In the current TTCN implementation of fl_TC_11_3_2a_Body(), replaced f_UTRAN34_ReleaseAM_RAB22_r7 with f_UTRAN34_ReleaseAM_RAB22_r7_NISPC (Please see screenshot)

	Module
	UTRAN34_NISPC.ttcn

	Label
	


Before:

	function fl_TC_11_3_2a_Body() runs on UTRAN_PTC
  {
    var PDPContextParameters v_PrimPDPContextParameters;
    var PDPContextParameters v_SecPDPContextParameters;
    var IP_VersionType v_IP_VersionType;
    var template (value) QualityOfService v_QualityOfService := f_UTRAN_InitialiseQoS(utran34_Cell1, cell_DCH_64kPS_RAB_SRB);
    var TI v_TI;
    var template (value) TrafficFlowTemplate v_TFT;
    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;
    var template (value) PacketFilter v_PacketFilterByRef;
    var integer v_PacketFilterLengthByref;
    var boolean v_IpTypeIsIPv4 := true;
    var integer v_LengthCount := 0;
    //var boolean v_GMM_DetachExpect := true; //@sic R5s141290 sic@
    timer t_T3395max := tsc_T3395max;
    //@siclog "Step 1-6" siclog@
    v_PrimPDPContextParameters := f_UTRAN34_ActivateFirstPDP_ContextNISPC(utran34_Cell1, v_QualityOfService, false, true); //check if 2 loops in the TC!
    v_IP_VersionType := v_PrimPDPContextParameters.ip_VersionType;
    if(v_IP_VersionType == iPv6){   //@sic R5s141153 sic@
        v_IpTypeIsIPv4 := false
    }
    //Generate traffic flow template
    f_Get_PacketFilter_10_9_1(v_PacketFilterByRef, v_PacketFilterLengthByref, v_IpTypeIsIPv4, 1);
    v_PacketFilterList[0]  := v_PacketFilterByRef;
    v_LengthCount := v_LengthCount + v_PacketFilterLengthByref + 1;
    v_TFT := cs_TrafficFlowTemplate(v_LengthCount, v_PacketFilterList);
    v_TFT.iei := '36'O;
    //@sic R5s140400 R5s140558 sic@
    //@siclog "Step 7-11" siclog@
    v_SecPDPContextParameters := f_UTRAN34_SecPDP_Context_MT_r7(utran34_Cell1, v_TFT, v_PrimPDPContextParameters, v_PrimPDPContextParameters.qOS, "Test Case 11.3.2a, Step 9");
    //@siclog "Step 12" siclog@
    //Wait for T3380 seconds to ensure no further activaate request message comes from UE
    f_UTRAN_CheckNoDataInd(tsc_T3380max, "Test Case 11.3.2a, Step 12");
    //@siclog "Step 13-14" siclog@
    //Request a deactivation of the primary PDP context and strarts waiting for T3395. SM cause set to #36
    v_TI := v_PrimPDPContextParameters.send_TI; //@sic R5s140400 sic@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3,  cs_U_DeactPdpContextReq(cs_DeactPdpContextReqMt(v_TI))));
    v_TI := v_PrimPDPContextParameters.received_TI; //@sic R5s140400 sic@
    fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3395max, cr_U_DeactPdpContextAcp(cr_DeactPdpContextAcpMo(v_TI)), "Test Case 11.3.2a, Step 14");
    //It is assumed that the RAB release is faster that the T3395max;
    //otherwise the timeout will be caught in the RRC default
    t_T3395max.start;
    //Be prepared to expect a detach while releasing the RAB
    //The UE may (optionally) send a Detach Request (only Note 1). The SS shall wait up to 'T3395' seconds for the Detach Request. While waiting the SS releases the RAB associated with this PDP Context. If the SS receives a Detach Request the test execution continues from step A15, otherwise from step B15. SS shall check that no further deactivate request messages are sent during this time.
    //@siclog "Step 14b" siclog@
    //SS releases RAB for primary PDP context
    f_UTRAN34_ReleaseAM_RAB20_NISPC(utran34_Cell1); //@sic R5s140400 sic@
    f_UTRAN34_ReleaseAM_RAB22_r7 (utran34_Cell1);   //@sic R5s150002, R5-150614 sic@

    alt{ //@sic R5s141290 sic@
        //Detach Request message is received
        [] U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                                    tsc_RB3,
                                                    cr_U_DetachRequest(cr_DetachTypeReAttNotRequiredGPRS, cdr_MobileIdTMSI_tlv(?, '24'O) ifpresent)))
          {
            t_T3395max.stop;
            //@siclog "Step A15-A16" siclog@
            fl_SendModifyPDPContReq(v_PrimPDPContextParameters, v_SecPDPContextParameters);
            //@siclog "Step A17" siclog@
            f_UTRAN_CheckNoDataInd(tsc_T3386max, "Test Case 11.3.2a, Step A17");
            //@siclog "Step A18" siclog@
            U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_DetachAcc));
          }
        //Detach Request message is not received
        [] t_T3395max.timeout
          {
            //@siclog "Step B15-B15A" siclog@
            fl_SendModifyPDPContReq(v_PrimPDPContextParameters, omit); //@sic R5s150002 sic@
            fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3386max, cr_U_SM_Status(cr_SM_StatusMO(cs_SM_Cause('51'O))), "Test Case 11.3.2a, Step B15A");
            //@siclog "Step B16-B17" siclog@
            fl_SendModifyPDPContReq(omit, v_SecPDPContextParameters); //@sic R5s150002 sic@
            fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3386max, cr_U_SM_Status(cr_SM_StatusMO(cs_SM_Cause('51'O))), "Test Case 11.3.2a, Step B17"); //@sic R5s140400 sic@
          }
    }
  } //end of the TC 11.3.2a


After:

	function fl_TC_11_3_2a_Body() runs on UTRAN_PTC
  {
    var PDPContextParameters v_PrimPDPContextParameters;
    var PDPContextParameters v_SecPDPContextParameters;
    var IP_VersionType v_IP_VersionType;
    var template (value) QualityOfService v_QualityOfService := f_UTRAN_InitialiseQoS(utran34_Cell1, cell_DCH_64kPS_RAB_SRB);
    var TI v_TI;
    var template (value) TrafficFlowTemplate v_TFT;
    var template (value) TrafficFlowTemplate.packetFilterList v_PacketFilterList;
    var template (value) PacketFilter v_PacketFilterByRef;
    var integer v_PacketFilterLengthByref;
    var boolean v_IpTypeIsIPv4 := true;
    var integer v_LengthCount := 0;
    //var boolean v_GMM_DetachExpect := true; //@sic R5s141290 sic@
    timer t_T3395max := tsc_T3395max;
    //@siclog "Step 1-6" siclog@
    v_PrimPDPContextParameters := f_UTRAN34_ActivateFirstPDP_ContextNISPC(utran34_Cell1, v_QualityOfService, false, true); //check if 2 loops in the TC!
    v_IP_VersionType := v_PrimPDPContextParameters.ip_VersionType;
    if(v_IP_VersionType == iPv6){   //@sic R5s141153 sic@
        v_IpTypeIsIPv4 := false
    }
    //Generate traffic flow template
    f_Get_PacketFilter_10_9_1(v_PacketFilterByRef, v_PacketFilterLengthByref, v_IpTypeIsIPv4, 1);
    v_PacketFilterList[0]  := v_PacketFilterByRef;
    v_LengthCount := v_LengthCount + v_PacketFilterLengthByref + 1;
    v_TFT := cs_TrafficFlowTemplate(v_LengthCount, v_PacketFilterList);
    v_TFT.iei := '36'O;
    //@sic R5s140400 R5s140558 sic@
    //@siclog "Step 7-11" siclog@
    v_SecPDPContextParameters := f_UTRAN34_SecPDP_Context_MT_r7(utran34_Cell1, v_TFT, v_PrimPDPContextParameters, v_PrimPDPContextParameters.qOS, "Test Case 11.3.2a, Step 9");
    //@siclog "Step 12" siclog@
    //Wait for T3380 seconds to ensure no further activaate request message comes from UE
    f_UTRAN_CheckNoDataInd(tsc_T3380max, "Test Case 11.3.2a, Step 12");
    //@siclog "Step 13-14" siclog@
    //Request a deactivation of the primary PDP context and strarts waiting for T3395. SM cause set to #36
    v_TI := v_PrimPDPContextParameters.send_TI; //@sic R5s140400 sic@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3,  cs_U_DeactPdpContextReq(cs_DeactPdpContextReqMt(v_TI))));
    v_TI := v_PrimPDPContextParameters.received_TI; //@sic R5s140400 sic@
    fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3395max, cr_U_DeactPdpContextAcp(cr_DeactPdpContextAcpMo(v_TI)), "Test Case 11.3.2a, Step 14");
    //It is assumed that the RAB release is faster that the T3395max;
    //otherwise the timeout will be caught in the RRC default
    t_T3395max.start;
    //Be prepared to expect a detach while releasing the RAB
    //The UE may (optionally) send a Detach Request (only Note 1). The SS shall wait up to 'T3395' seconds for the Detach Request. While waiting the SS releases the RAB associated with this PDP Context. If the SS receives a Detach Request the test execution continues from step A15, otherwise from step B15. SS shall check that no further deactivate request messages are sent during this time.
    //@siclog "Step 14b" siclog@
    //SS releases RAB for primary PDP context
    f_UTRAN34_ReleaseAM_RAB20_NISPC(utran34_Cell1); //@sic R5s140400 sic@
    //WA#11_3_2a f_UTRAN34_ReleaseAM_RAB22_r7 (utran34_Cell1);   //@sic R5s150002, R5-150614 sic@
    f_UTRAN34_ReleaseAM_RAB22_r7_NISPC (utran34_Cell1);//WA#11_3_2a
    alt{ //@sic R5s141290 sic@
        //Detach Request message is received
        [] U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                                    tsc_RB3,
                                                    cr_U_DetachRequest(cr_DetachTypeReAttNotRequiredGPRS, cdr_MobileIdTMSI_tlv(?, '24'O) ifpresent)))
          {
            t_T3395max.stop;
            //@siclog "Step A15-A16" siclog@
            fl_SendModifyPDPContReq(v_PrimPDPContextParameters, v_SecPDPContextParameters);
            //@siclog "Step A17" siclog@
            f_UTRAN_CheckNoDataInd(tsc_T3386max, "Test Case 11.3.2a, Step A17");
            //@siclog "Step A18" siclog@
            U_Dc.send(cas_DataReq_PS(utran_CellDedicated, tsc_RB3, cs_U_DetachAcc));
          }
        //Detach Request message is not received
        [] t_T3395max.timeout
          {
            //@siclog "Step B15-B15A" siclog@
            fl_SendModifyPDPContReq(v_PrimPDPContextParameters, omit); //@sic R5s150002 sic@
            fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3386max, cr_U_SM_Status(cr_SM_StatusMO(cs_SM_Cause('51'O))), "Test Case 11.3.2a, Step B15A");
            //@siclog "Step B16-B17" siclog@
            fl_SendModifyPDPContReq(omit, v_SecPDPContextParameters); //@sic R5s150002 sic@
            fl_NISPC_WaitForPS_UplinkDirectTransfer(tsc_T3386max, cr_U_SM_Status(cr_SM_StatusMO(cs_SM_Cause('51'O))), "Test Case 11.3.2a, Step B17"); //@sic R5s140400 sic@
          }
    }
  } //end of the TC 11.3.2a


